
Machine-Level Programming III:
Procedures

Sept 19, 2000

Topics
• IA32 stack discip line
• Register saving conv entions
• Creating pointers to loca l

variables
• Stack buffer overflow expl oits

– finger
– AIM (AOL Instant Messenger)

“The course that gives CMU its Zip!”

IA32 Stack
• Region of memory man aged with

stack discipline
• Register %esp indicates lowest

allocated position in stack
– i.e., address of top element

Pushing
• pushl Src
• Fetch operand at Src
• Decrement %esp by 4
• Write operand at addres s given

by %esp

Popping
• popl Dest
• Read operand at addres s given

by %esp

• Increment %esp by 4
• Write to Dest

Stack
Pointer

%esp

Stack Grows
Down

Increasing
Addresses

Stack “Top”

Stack “Bottom”

0x108

0x10c

0x110

0x104

555

%esp

213%eax

213

123

0x104

%edx

Stack Operation Examples

0x108

0x10c

0x110

555

%esp

213%eax

123

0x108

%edx

pushl %eax

0x108

0x10c

0x110

213

%esp

213%eax

123

0x108

%edx

popl  %edx

Procedure Control Flow

Use stack to support procedure call a nd return

Procedure call:
call label Push return address on stack; Jump to label

Return address value
• Address of instruction be yond call

• Example from disas sembly
 804854e: e8 3d 06 00 00 call   8048b90 <mai n>

 8048553: 50             pushl   %eax

– Return address = 0x8048553

Procedure return:
• ret Pop address from stack; Jump to address

0x108

0x10c

0x110

0x104

0x804854e

0x8048553

123

Procedure Call / Return Example

0x108

0x10c

0x110

%esp

%eip

123

0x108

call   8048b90

0x108

0x10c

0x110

123

ret

804854e: e8 3d 06 00 00 call   8048b90 <m ain>
8048553: 50             pushl   % eax

0x8048b90

0x104%esp

%eip 0x8048553

0x108%esp

%eip

%eip is program counter

Stack-Based Languages
Languages that Support Recursion

• e.g., C, Pascal, Ja va
• Code must be “ Reentrant ”

– Multiple simultaneous instantiations of single procedure
• Need some place to store state of each instantiation

– Arguments

– Local variables
– Return pointer

Stack Discipline
• State for given procedure needed for limited tim e

– From when called to when return

• Callee returns before caller d oes

Stack Allocated in Frames
• state for single proce dure instantiation

Call Chain Example
Code Structure

yoo (…)
{

•
•
who();
•
•

}

who(…)
{

•
•
amI ();
•
•

}

amI (…)
{

•
•
amI ();
•
•

}

yoo

who

amI

amI

amI

Call Chain

• Procedure  amI  recursive

Stack
Pointer

%esp

yoo

who

amI

amI

amI

•
•
•

Frame
Pointer

%ebp

Stack Grows

Increasing
Addresses

Stack
“Top”

IA32 Stack Structure
Stack Growth

• Toward lower addresses

Stack Pointer
• Address of next avai lable

location in stack
• Use register %esp

Frame Pointer
• Start of current stack frame
• Use register %ebp

IA32/Linux  Stack Frame
Callee Stack Frame (“Top” to

Bottom)
• Parameters for called functions
• Local variables

– If can’t keep in registers

• Saved register contex t
• Old frame pointer

Caller Stack Frame
• Return address

– Pushed by call  instruction

• Arguments for this call

Stack Pointer
(%esp)

Frame Pointer
(%ebp)

Return Addr

Saved
Registers

Argument
Build

Old %ebp

Local
Variables

Arguments

Caller
Frame

Revisiting swap

void swap( int  *xp , int  *yp)
{
  int  t0 = * xp ;
  int  t1 = * yp ;
  * xp = t1;
  * yp = t0;
}

int  zip1 = 15213;
int  zip2 = 91125;

void call_swap()
{
  swap(&zip1, &zi p2);
}

call_swap:
• • •
pushl  $zip2
pushl  $zip1
call swap
• • •

&zip2

&zip1

Rtn adr %esp

Resulting
Stack

•
•
•

Revisiting swap

void swap( int  *xp , int  *yp)
{
  int  t0 = * xp ;
  int  t1 = * yp ;
  * xp = t1;
  * yp = t0;
}

swap:
pushl  %ebp
movl  %esp ,%ebp
pushl  %ebx

movl  12(% ebp),% ecx
movl  8(% ebp),% edx
movl  (%ecx ),% eax
movl  (%edx ),% ebx
movl  %eax ,(%edx )
movl  %ebx ,(%ecx )

movl  -4(% ebp),% ebx
movl  %ebp ,%esp
popl  %ebp
ret

Body

Set
Up

Finish

CS 213 F’00– 12 –class07. ppt

swap Setup

swap:
pushl  %ebp
movl  %esp ,%ebp
pushl  %ebx

yp

xp

Rtn adr

Old %ebp %ebp 0 

 4 

 8 

12 

Offset

Resulting
Stack

•
•
•

&zip2

&zip1

Rtn adr %esp

Entering
Stack

•
•
•

%ebp

Old %ebx %esp

swap Finish

movl  -4(% ebp),% ebx
movl  %ebp ,%esp
popl  %ebp
ret

yp

xp

Rtn adr

Old %ebp %ebp 0 

 4 

 8 

12 

Offset

swap’s
Stack

•
•
•

&zip2

&zip1 %esp

Exiting
Stack

•
•
•

%ebp

Old %ebx %esp-4 

Observation
• Saved & restored registe r %ebx

• Didn’t do so for %eax, %ecx, or %edx

Register Saving Conventions
When procedure  yoo  calls who:

•  yoo  is the caller , who is the  callee

Can Register be Used for Temporary Storage?

• Contents of register %edx overwritten by who

Conventions
• “Caller Save”

– Caller saves temporary in its frame before calling

• “Callee  Save”
– Callee saves temporary in its frame before using

yoo :
• • •
movl  $15213, % edx
call who
addl  %edx , %eax
• • •
ret

who:
• • •
movl 8(%ebp), % edx
addl  $91125, % edx
• • •
ret

IA32/Linux Register Usage
• Surmised by looking at

code examples

Integer Registers
• Two have special u ses

%ebp, %esp

• Three managed as callee -
save
%ebx, %esi , %edi

– Old values saved on
stack prior to using

• Three managed as caller-
save
%eax, %edx, %ecx

– Do what you please, but
expect any callee to do
so, as well

• Register %eax also stores
returned value

%eax

%edx

%ecx

%ebx

%esi

%edi

%esp

%ebp

Caller-Save
Temporaries

Callee-Save
Temporaries

Special

int rfact (int  x)
{
  int rval ;
  if (x <= 1)
    return 1;
  rval  =  rfact (x-1);
  return  rval  * x;
}

.globl rfact
.type

rfact ,@function
rfact :

pushl  %ebp
movl  %esp ,%ebp
pushl  %ebx
movl  8(% ebp),% ebx
cmpl  $1,% ebx
jle  .L78
leal  -1(% ebx ),%eax
pushl  %eax
call  rfact
imull  %ebx,% eax
jmp  .L79
.align 4

.L78:
movl  $1,% eax

.L79:
movl  -4(% ebp),%ebx
movl  %ebp ,%esp
popl  %ebp
ret

Recursive Factorial

Complete Assembly
• Assembler directives

– Lines beginning with “. ”

– Not of concern to us
• Labels

–  .Lxx

• Actual instructions

Rfact  Stack Setup

rfact :
pushl  %ebp
movl  %esp ,%ebp
pushl  %ebx

Entering Stack

x

Rtn adr

Old %ebp %ebp 0 

 4 

 8 

Old %ebx %esp-4 

Caller

Callee

x

Rtn adr

Caller

%esp

Rfact  Body

Registers
$ebx Stored value of x
$eax

– Temporary value of x-1

– Returned value from rfact (x-1)

– Returned value from this call

movl  8(% ebp),% ebx # ebx = x
cmpl  $1,% ebx # Compare x : 1
jle  .L78 # If <= goto Term
leal  -1(% ebx),% eax # eax = x-1
pushl  %eax # Push x-1
call rfact # rfact (x-1)
imull  %ebx ,%eax # rval * x
jmp  .L79 # Goto done

.L78: # Term:
movl  $1,% eax # return val = 1

.L79: # Done:

int rfact (int  x)
{
  int rval ;
  if (x <= 1)
    return 1;
  rval  =  rfact (x-1);
  return  rval  * x;
}

Rfact  Recursion

pushl  %eax

x

Rtn adr

Old %ebp %ebp

Old %ebx

%espx-1

x-1%eax

x%ebx

x

Rtn adr

Old %ebp %ebp

Old %ebx %esp

x-1%eax

x%ebx

leal  -1(% ebx),% eax

x

Rtn adr

Old %ebp %ebp

Old %ebx

%esp

x-1

Rtn adr

x-1%eax

x%ebx

call rfact

Rfact  Result

imull %ebx,%eax

x

Rtn adr

Old %ebp %ebp

Old %ebx

%espx-1

x!%eax

x%ebx

x

Rtn adr

Old %ebp %ebp

Old %ebx

%espx-1

(x-1)!%eax

x%ebx

Return from Call

Rfact  Completion

movl  -4(% ebp),% ebx
movl  %ebp ,%esp
popl  %ebp
ret

x

Rtn adr

Old %ebp %ebp 0 

 4 

 8 

Old %ebx

%esp

-4 

x %esp

x!%eax

Old %ebx%ebx

x!%eax

x%ebx

x-1-8 

Pointer Code

void s_helper
  ( int x,  int  *accum )
{
  if (x <= 1)
    return;
  else {
    int z = * accum * x;
    * accum = z;
    s_helper (x-1 ,accum );
  }
}

int sfact (int  x)
{
  int val  = 1;
  s_helper(x, & val );
  return  val;
}

Top-Level CallRecursive Procedure

• Pass pointer to update location
• Uses tail recursion

– But GCC only partially optimizes it

Creating & Initializing Pointer

int sfact (int  x)
{
  int val  = 1;
  s_helper(x, &val);
  return  val;
}

_sfact :
pushl  %ebp # Save % ebp
movl  %esp,% ebp # Set % ebp
subl  $16,% esp # Add 16 bytes
movl  8(% ebp),%edx # edx = x
movl  $1,-4(% ebp) # val = 1

Using Stack for Local
Variable
• Variable  val  must be stored

on stack
– Need to create pointer to it

• Compute pointer as -4(% ebp)

• Push on stack as second
argument

Initial part of sfact

x

Rtn adr

Old %ebp %ebp  0 

  4 

  8 

val = 1

%esp

 -4 

Unused-12 

 -8 

-16

Passing Pointer

int sfact (int  x)
{
  int val  = 1;
  s_helper(x, &val);
  return  val;
}

leal  -4(% ebp),% eax # Compute & val
pushl  %eax # Push on stack
pushl  %edx # Push x
call _s_helper # call
movl  -4(% ebp),% eax # Return val
• • • # Finish

Calling s_helper  from sfact

x

Rtn adr

Old %ebp %ebp  0 

  4 

  8 

val = 1

%esp

 -4 

Unused-12 

 -8 

-16

&val

x

Stack at time of ca ll

Using Pointer

• • •
movl  %ecx,% eax # z = x
imull  (%edx ),%eax # z *= * accum
movl  %eax,(% edx ) # * accum = z
• • •

void s_helper
  ( int x, int  *accum )
{
  • • •
    int  z = * accum * x;
    * accum = z;
  • • •
}

• Register %ecx holds x

• Register %edx holds accum

– Use access (%edx )  to reference memory

Internet worm and IM War
November, 1988

• Internet Worm attacks thousands of Internet hos ts.
• How did it happen?

July, 1999
• Microsoft launches MSN Messenge r (instant messaging system).
• Messenger clients ca n access popular AOL In stant Messaging

Service (AIM) servers

AIM
server

AIM
client

AIM
client

MSN
client

MSN
server

CS 213 F’00– 27 –class07. ppt

Internet Worm and IM War ( cont)
August 1999

• Mysteriously, Messenger c lients can no longe r access AIM servers.
• Even though the AIM protocol i s an open, publishe d standard.
• Microsoft and AOL begin the IM war:

– AOL changes server to disallow Messenger clients
– Microsoft makes changes to clients to defeat AOL changes.

– At least 13 such skirmishes.
• How did it happen?

The Internet Worm and AOL/Microsoft War were both
based on stack buffer overflow  exploits!

– many Unix functions, such as gets() and strcpy(), do not check
argument sizes.

– allows target buffers to overflow.

Stack buffer overflows

void bar() {
  char buf [64];
  gets( buf );
  ...
}

void foo (){
  bar();
  ...
}

Stack 
before call to gets()

A

Old %ebp

buf

return
address

A

foo stack frame

bar stack frame

Stack buffer overflows (cont)

void bar() {
  char buf [64];
  gets( buf );
  ...
}

void foo (){
  bar();
  ...
}

Stack 
after call to gets()

B

return
address

A

foo stack frame

bar stack frame

B

exploit
code

pad

When bar() returns, control p asses silently to B instead of A!!

data 
written

by
gets()

Exploits based on buffer overflows
Buffer overflow bugs allow remote machine s to

execute arbitrary code on victim machines.
Internet worm

• Early versions of the finger server ( fingerd) used gets()  to read the
argument sent by the client:
– finger  droh @cs.cmu .edu

• Worm attacked fingerd client by sending ph ony argument:
– finger “exploit c ode  padding  new re turn address”

– exploit code: exe cuted a root shell on th e victim machine with a
direct TCP connection to the attacker.

IM War
• AOL exploited existing buffer overflow bug in AIM clients
• exploit code: returned 4-byte signature (the by tes at some locati on in

the AIM client) to server.
• When Microsoft changed c ode to match signatu re, AOL changed

signature location.

Main Ideas
Stack Provides Storage for Procedure Instantiation

• Save state
• Local variables
• Any variable for which mu st create pointer

Assembly Code Must Manage Stack
• Allocate /  deallocate  by decrementing  / incrementing sta ck pointer
• Saving / restoring regis ter state

Stack Adequate for All Forms of Recursion
• Including multi-way an d mutual recursion exa mples in the bonus

slides.

Good programmers know the stack dis cipline and are
aware of the dangers of stack buffer overflow s.

Free Bonus Slides!
(not covered in lectu re)

Topics
• how the stack supports multi-

way recursion.
• how the stack supports mutual

recursion.

class07. ppt

Multi-Way Recursion

int bfact (int  x)
{
  return r_prod(1 ,x);
}

int  r_prod
  ( int from, int to)
{
  int  middle;
  int prodA , prodB ;
  if (from >= to)
    return from;
  middle = (from + to) >> 1;
  prodA  = r_prod(from, middle);
  prodB  = r_prod(middle+1, to);
  return  prodA  * prodB ;
}

• Compute product x * (x+1) * … * (y–1) * y
• Split into two ranges:

– Left: x * (x+1) * … * (m–1) * m
– Right: (m+1) * … * (y–1) * y

m = (x+y)/2
• No real advantage a lgorithmically

Top-Level Call

Binary Splitting Example

bfact (6)

r_prod(1,6)

r_prod(1,3) r_prod(4,6)

r_prod(1,2) r_prod(3,3) r_prod(4,5) r_prod(6,6)

r_prod(1,1) r_prod(2,2) r_prod(4,4) r_prod(5,5)

1 2 4 5

2

6 120

720

6203

Multi-Way Recursive Code
_r_prod:

 • • • # Setup
movl  8(% ebp),% eax # eax = from
movl  12(% ebp),%edi # edi = to
cmpl  %edi ,%eax # from : to
jge  L8 # if >= goto done
leal  (% edi,% eax),% ebx # from + to
sarl  $1,% ebx # middle
pushl  %ebx # 2nd arg : middle
pushl  %eax # 1st arg : from
call _r_prod # 1st call
pushl  %edi # 2nd arg : to
movl  %eax ,%esi # esi = ProdA
incl  %ebx # middle + 1
pushl  %ebx # ... 1st arg
call _r_prod # 2nd call
imull  %eax,% esi # ProdA * ProdB
movl  %esi ,%eax # Return value

L8: # done:
 • • • # Finish

Stack Frame

$eax

from

return values

Callee Save Regs.
$ebx middle

$edi to

$esi  prodA

from

to

Rtn Adr
Old $ebp

Old $edi

Old $esi

Old $ebx

 12

  8

  4

  0

 -4

 -8

-12

L8: # done:
leal  -12(% ebp),%esp # Set Stack Ptr
popl  %ebx # Restore % ebx
popl  %esi # Restore % esi
popl  %edi # Restore % edi
movl  %ebp ,%esp # Restore % esp
popl  %ebp # Restore % ebp
ret # Return

from

to

Rtn Adr
Old $ebp

Old $edi

Old $esi

Old $ebx

 12

  8

  4

  0

 -4

 -8

-12

Multi-Way Recursive Code Finish

Stack
• After making recursive calls, still has two a rguments on top

Finishing Code
• Moves stack pointer to start of saved register a rea
• Pops registers

Arg 2
Arg 1

-16

-20

Mutual Recursion

int lrfact (int  x)
{
  int  left = 1;
  return
   left_prod(&lef t, &x);
}

int left_prod
(int * leftp, int  *rightp )
{
  int  left = *leftp ;
  if (left >= * rightp)
    return left;
  else {
    int  plus1 = left+1;
    return left *
      right_prod(&plus1,  rightp);
  }
}

int right_prod
(int * leftp, int  *rightp )
{
  int  right = * rightp;
  if (* leftp == right)
    return right;
  else {
    int  minus1 = right-1;
    return right *
      left_prod(leftp, &minus1);
  }
}

Top-Level Call

Mutually Recursive Execution Example
Calling

• Recursive routines pa ss two
arguments
– Pointer to own local variable
– Pointer to caller’s local

variable

lrfact ( 4 )

1left:

4x:

left_prod(  ,  )

2plus1:

right_prod(  ,  )

3minus1:

left_prod(  ,  )

3plus1:

right_prod(  ,  )
3

6

24

24

24

Implementation of  lrfact

  int  left = 1;
  return left_pro d(&left, &x);

leal  8(% ebp),%edx # edx = &x
pushl  %edx # push &x
leal  -4(% ebp),% eax # eax = &left
pushl  %eax # push &left
call _left_prod # Call

Call to Recursive Routine

Code for Call

x

Rtn adr

Old %ebp %ebp  0 

  4 

  8 

left= 1

%esp

 -4 

Unused-12 

 -8 

-16

&x

&left

Stack at time of ca ll

Implementation of left_prod

 int  plus1 = left+1;
 return left *
   right_prod(&pl us1, rightp );

Call to Recursive Routine

# %ebx  holds left
# %edx holds rightp
leal  1(% ebx ),%ecx # left+1
movl  %ecx,-4(% ebp) # Store in plus1
pushl  %edx # Push  rightp
leal  -4(% ebp),% eax # &plus1
pushl  %eax # Push &plus1
call _right_pro d # Call

leftp

Rtn adr

Old %ebp %ebp  0 

  4 

  8 

plus1

%esp

 -4 

Unused-12 

 -8 

-16

rightp

&plus1

Stack at time of ca ll

rightp 12 

Tail Recursion

int tfact (int  x)
{
  return t_helper (x, 1);
}

t_helper(x,  val )
{
   • • •
  return
   t_helper( Xexpr , Vexpr )
}

General Form

Form
• Directly return value returne d

by recursive call

Consequence
• Can convert into loop

int  t_helper
  ( int x, int val )
{
  if (x <= 1)
    return  val ;
  return
    t_helper(x-1,  val*x);
}

Tail Recursive Procedure

Top-Level Call

Removing Tail Recursion

int  t_helper
  ( int x, int val )
{
 start:
  if (x <= 1)
    return  val ;
  val  = val *x;
  x = x-1;
  goto  start;
}

t_helper(x,  val )
{
 start:
   • • •
  val  = Vexpr ;
  x =  Xexpr ;
  goto  start;
}

Optimized General Form Resulting Code

Effect of Optimization
• Turn recursive chain into single procedure
• No stack frame need ed
• Constant space require ment

– Vs. linear for recursive version

Generated Code for Tail Recursive Proc.

# %edx = x
# %ecx =  val

L53: # start:
cmpl  $1,% edx # x : 1
jle  L52 # if <= goto done
movl  %edx ,%eax # eax = x
imull  %ecx,% eax # eax = val * x
decl  %edx # x--
movl  %eax ,%ecx # val = val * x
jmp  L53 # goto start

L52: # done:

Optimized Form

int  t_helper
  ( int x,  int val )
{
 start:
  if (x <= 1)
    return  val ;
  val  = val *x;
  x = x-1;
  goto  start;
}

Code for Loop

Registers
$edx x

$ecx val


